




UNIVERSITÀ DELLA CALABRIA



Introduzione alle funzionalità del pacchetto

IBM ILOG CPLEX Optimization Studio

A cura di:

Cristian Belfiore, Dottore Magistrale in Ingegneria Gestionale

de-Health Lab - Laboratory of Decision Engineering for Health Care Services

Ponte Pietro Bucci 41C, 8° piano - 87036 Rende (Cosenza)

mail: cristian.belfiore@unical.it



UNIVERSITÀ DELLA CALABRIA 
DIPARTIMENTO DI
INGEGNERIA MECCANICA,
ENERGETICA E GESTIONALE
DIMEG



IBM ILOG Script

Alcune operazioni necessarie per la costruzione e la risoluzione di un modello di ottimizzazione **non possono essere rappresentate attraverso il paradigma dichiarativo dal linguaggio OPL standard.**

IBM ILOG Script

(linguaggio procedurale incorporato in OPL)



IBM ILOG Script

Alcune operazioni necessarie per la costruzione e la risoluzione di un modello di ottimizzazione **non possono essere rappresentate attraverso il paradigma dichiarativo dal linguaggio OPL standard.**

IBM ILOG Script

(linguaggio procedurale incorporato in OPL)

IBM ILOG Script estende OPL e permette:

- preprocessamento dei dati di input
- postprocessamento dei risultati
- settaggio dei parametri di funzionamento di CPLEX
- chiamata a programmi o porzioni di codice esterni
- molto altro...



IBM ILOG Script

Alcune operazioni necessarie per la costruzione e la risoluzione di un modello di ottimizzazione **non possono essere rappresentate attraverso il paradigma dichiarativo dal linguaggio OPL standard.**

IBM ILOG Script

(linguaggio procedurale incorporato in OPL)

ATTENZIONE:

Codice che definisce il modello di ottimizzazione (OPL) → elementi elaborati **simultaneamente** nel processo di soluzione

Righe di codice dello Script → elaborate rigorosamente **in sequenza** (linguaggio procedurale)



IBM ILOG Script: implementare una euristica

RideEasy ha deciso di lanciare una bicicletta in edizione speciale di alta qualità destinata al mercato di New York. Tali biciclette, denominate CycleMe, sono realizzate su ordinazione in una officina dedicata alla periferia della città. La responsabile dell'officina ha appena ricevuto gli ordinativi da consegnare nei prossimi 20 giorni (t); indagini di mercato hanno mostrato che il grado di soddisfazione del cliente risente, oltre che dalla qualità costruttiva delle biciclette, dalla puntualità nelle consegne.

La responsabile della produzione deve definire il piano di produzione tenendo conto

delle **capacità giornaliera di produzione (c_t)**, variabile nei prossimi 20 giorni

e cercando di minimizzare i costi complessivi, suddivisibili nelle seguenti voci (tutte variabili nei prossimi 20 giorni):

costi di setup (s_t): costi fissi, sostenuti giornalmente, se in quel giorno vengono prodotte biciclette

costi di produzione (p_t): costo di produzione di ogni bicicletta prodotta

costi di stoccaggio (h_t): costo di stoccaggio per unità di tempo e di prodotto

* da:

Nickel S., Steinhardt C., Schlenker H., Burkart W., "Decision Optimization with IBM ILOG CPLEX Optimization Studio: A Hands-On Introduction to Modeling with the Optimization Programming Language (OPL)", Springer Nature, 2022, P. 257

IBM ILOG Script: implementare una euristica

```

1 /*****
2  * OPL 22.1.0.0 Model
3  * Author: Win10
4  * Creation Date: 30 ago 2023 at 11:11:15
5  Dal libro di Nickel et al. pag. 257
6  *****/
7
8 int numPeriodi = ...;
9 range Periodi = 1..numPeriodi;
10
11 int C_SetUp [Periodi] = ...;
12 int C_Stoccaggio [Periodi] = ...;
13 int C_Produzione [Periodi] = ...;
14 int Domanda [Periodi] = ...;
15 int Capacita [Periodi] = ...;
16
17 dvar boolean z[Periodi];
18 dvar float+ i[0..numPeriodi];
19 dvar float+ q[Periodi];
20
21 minimize sum (t in Periodi) (C_SetUp[t]*z[t] + C_Stoccaggio[t]*i[t] + C_Produzione[t]*q[t]);
22
23 subject to{
24
25     i[0] == 0;
26
27 forall (t in Periodi)
28     i[t] == i[t-1] + q[t] - Domanda[t];
29
30 forall (t in Periodi)
31     q[t] <= Capacita[t]*z[t];
32 }

```

Table 16.1 Data for lot sizing [with d_t for demand on day t , c_t for available capacities for production on day t , s_t for setup costs on day t in USD, h_t for holding costs on day t in USD, p_t for production costs on day t in USD]

t	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20
d_t	12	6	20	5	7	25	0	5	5	21	9	5	27	0	5	20	5	10	5	25
c_t	20	15	5	30	20	20	21	10	10	5	0	12	20	5	20	5	40	20	10	20
s_t	25	30	40	20	80	80	150	30	45	60	30	90	80	150	25	20	30	40	20	80
h_t	6	5	4	5	6	9	9	6	5	4	5	6	9	9	6	5	4	5	6	9
p_t	5	3	7	2	1	10	15	2	1	5	1	10	15	2	1	10	1	5	1	10

$$\text{Minimize } \sum_{t=1}^T (s_t \cdot z_t + h_t \cdot i_t + p_t \cdot q_t) \quad (16.1)$$

subject to the constraints

$$i_0 = 0 \quad (16.2)$$

$$i_t = i_{t-1} + q_t - d_t \quad \forall t = 1, \dots, T \quad (16.3)$$

$$q_t \leq c_t \cdot z_t \quad \forall t = 1, \dots, T \quad (16.4)$$

$$q_t, i_t \geq 0 \quad \forall t = 1, \dots, T \quad (16.5)$$

$$z_t \in \{0; 1\} \quad \forall t = 1, \dots, T \quad (16.6)$$



IBM ILOG Script: implementare una euristica

The screenshot displays the IBM ILOG Script interface. The top pane shows the 'LotSizing_Model.mod' file with the following code:

```
1 /*****  
2 * OPL 22.1.0.0 Data  
3 * Author: Win10  
4 * Creation Date: 30 ago 2023 at 11:11:15  
5 *****/  
6  
7 numPeriodi = 20;  
8  
9 C_SetUp      = [25, 30, 40, 20, 80, 80, 150, 30, 45, 60, 30, 90, 80, 150, 25, 20, 30, 40, 20, 80];  
10 C_Stoccaggio = [6, 5, 4, 5, 6, 9, 9, 6, 5, 4, 5, 6, 9, 9, 6, 5, 4, 5, 6, 9];  
11 C_Produzione = [5, 3, 7, 2, 1, 10, 15, 2, 1, 5, 1, 10, 15, 2, 1, 10, 1, 5, 1, 10];  
12 Domanda     = [12, 6, 20, 5, 7, 25, 0, 5, 5, 21, 9, 5, 27, 0, 5, 20, 5, 10, 5, 25];  
13 Capacita    = [20, 15, 5, 30, 20, 20, 21, 10, 10, 5, 0, 12, 20, 5, 20, 5, 40, 20, 10, 20];  
14
```

The bottom pane shows the 'Statistics' tab with the following output:

```
// solution (optimal) with objective 2958  
// Quality Incumbent solution:  
// MILP objective                2,9580000000e+03  
// MILP solution norm |x| (Total, Max)  4,02000e+02  2,50000e+01  
// MILP solution error (Ax=b) (Total, Max)  0,00000e+00  0,00000e+00  
// MILP x bound error (Total, Max)  0,00000e+00  0,00000e+00  
// MILP x integrality error (Total, Max)  0,00000e+00  0,00000e+00  
// MILP slack bound error (Total, Max)  0,00000e+00  0,00000e+00  
//  
z = [1  
      1 1 1 1 1 0 1 1 1 0 1 1 0 1 1 1 0 1 1];  
i = [0 6 15 0 7 20 15 15 20 25 9 0 7 0 0 15 0 10 0 5 0];  
q = [18 15 5 12 20 20 0 10 10 5 0 12 20 0 20 5 15 0 10 20];
```

Two red circles highlight specific elements: the 'Statistics' tab in the bottom pane and the '00:00:01:50' timer in the bottom right corner.



IBM ILOG Script: implementare una euristica

LP-and-Fix heuristic (Pochet and Wolsey, 2006)

- 1) Risolvi il rilassato lineare del problema MIP di partenza ottenuto sostituendo le variabili $z_t \in \{0, 1\}$ con le variabili $z_t \in [0, 1]$.
- 2) Fissa tutte le variabili che assumono spontaneamente valore 0 o 1 nella soluzione ottima del rilassato lineare ottenuto al passo precedente.
- 3) Risolvi il problema MIP di partenza (considerando le variabili fissate).



IBM ILOG Script: implementare una euristica

```
1  /*****
2  * OPL 22.1.0.0 Model
3  * Author: Win10
4  * Creation Date: 30 ago 2023 at 11:33:35
5  *****/
6  main {
7    // Creazione del modello
8    var source = new IloOplModelSource("LotSizing_Model.mod");
9    var def    = new IloOplModelDefinition(source);
10   var opl    = new IloOplModel(def, cplex);
11
12   // Aggiunta dei dati sorgente
13   var data   = new IloOplDataSource("LotSizing_Data.dat");
14   opl.addDataSource(data)
15
16   // Rilassamento dei vincoli di integrità sulle variabili intere
17   opl.convertAllIntVars();
18
19   // Generazione del modello
20   opl.generate();
21
22   if (cplex.solve()) {
23     // Creazione e generazione di una seconda istanza del modello
24     var cplex2 = new IloCplex();
25     var opl2   = new IloOplModel(def, cplex2);
26     opl2.addDataSource(data);
27     opl2.generate();
28   }
29
30   for (var t=1; t <= opl.dataElements.numPeriodi; t++){
31     if (opl.z[t].solutionValue == 0 || opl.z[t].solutionValue == 1){
32       opl2.z[t].LB = opl.z[t].solutionValue;
33       opl2.z[t].UB = opl.z[t].solutionValue;
34     }
35   }
36
37   if (cplex2.solve()) {
38     writeln ("LP-and-Fix-Heuristic");
39     writeln ();
40     writeln ("Days of production: " + opl2.z.solutionValue);
41     writeln ("Inventories: "      + opl2.i.solutionValue);
42     writeln ("Lot sizes: "        + opl2.q.solutionValue);
43     writeln ("Total cost: "       + cplex2.getObjValue());
44   }
45 }
46 }
```



IBM ILOG Script: implementare una euristica

```
>> Scripting log (drop script code here to execute it)
LP-and-Fix-Heuristic

Days of production: [1 1 1 1 1 0 0 1 1 1 0 1 1 0 1 1 1 0 1 1]
Inventories: [0 8 17 2 27 40 15 15 20 25 9 0 7 0 0 15 0 10 0 5 0]
Lot sizes: [20 15 5 30 20 0 0 10 10 5 0 12 20 0 20 5 15 0 10 20]
Total cost: 2974
```

Writable	Insert	1:1:0-0	00:00:00:73
----------	--------	---------	-------------



IBM ILOG Script: implementare una euristica

Modified LP-and-Fix heuristic

- 1) Risolvi il rilassato lineare del problema MIP di partenza ottenuto sostituendo le variabili $z_t \in \{0, 1\}$ con le variabili $z_t \in [0, 1]$.
- 2) Fissa a 0 tutte le variabili che assumono spontaneamente valore 0 nella soluzione ottima del problema rilassato e fissa invece ad 1 tutte quelle che, in questa, assumono valore maggiore di 0.5.
- 3) Risolvi il problema MIP di partenza (considerando le variabili fissate).



IBM ILOG Script: implementare una euristica

```
H_1.2.mod X
4  * Creation Date: 30 ago 2023 at 13:02:35
5  *****/
6 main {
7  // Creazione del modello
8  var source = new IloOplModelSource("LotSizing_Model.mod");
9  var def    = new IloOplModelDefinition(source);
10 var opl    = new IloOplModel(def, cplex);
11
12 // Aggiunta dei dati sorgente
13 var data   = new IloOplDataSource("LotSizing_Data.dat");
14 opl.addDataSource(data)
15
16 // Rilassamento dei vincoli di integrità sulle variabili intere
17 opl.convertAllIntVars();
18
19 // Generazione del modello
20 opl.generate();
21
22 if (cplex.solve()) {
23 // Creazione e generazione di una seconda istanza del modello
24 var cplex2 = new IloCplex();
25 var opl2   = new IloOplModel(def, cplex2);
26 opl2.addDataSource(data);
27 opl2.generate();
28 }
29
30 for (var t=1; t <= opl.dataElements.numPeriodi; t++){
31   if (opl.z[t].solutionValue == 0){
32     opl2.z[t].LB = opl.z[t].solutionValue;
33     opl2.z[t].UB = opl.z[t].solutionValue;
34   }
35   else if (opl.z[t].solutionValue > 0.5){
36     opl2.z[t].LB = 1;
37     opl2.z[t].UB = 1;
38   }
39 }
40
41 if (cplex2.solve()) {
42   writeln ("LP-and-Fix-Heuristic");
43   writeln ();
44   writeln ("Days of production: " + opl2.z.solutionValue);
45   writeln ("Inventories: "      + opl2.i.solutionValue);
46   writeln ("Lot sizes: "       + opl2.q.solutionValue);
47   writeln ("Total cost: "      + cplex2.getObjValue());
48 }
49 }
```



IBM ILOG Script: implementare una euristica


```
>> Scripting log (drop script code here to execute it)
LP-and-Fix-Heuristic

Days of production: [1 1 1 1 1 0 1 1 1 1 0 1 1 0 1 1 1 0 1 1]
Inventories: [0 6 15 0 12 25 0 15 20 25 9 0 7 0 0 15 0 10 0 5 0]
Lot sizes: [18 15 5 17 20 0 15 10 10 5 0 12 20 0 20 5 15 0 10 20]
Total cost: 2983
```

Writable	Insert	31 : 39 : 983 - 983	00:00:00:81
----------	--------	---------------------	-------------



IBM ILOG Script: implementare una euristica






Riferimenti:

- “IBM ILOG CPLEX Optimization Studio OPL Language User’s Manual”, Version 12 Release 8, International Business Machines Corporation.
- Nickel S., Steinhardt C., Schlenker H., Burkart W., “Decision Optimization with IBM ILOG CPLEX Optimization Studio: A Hands-On Introduction to Modeling with the Optimization Programming Language (OPL)”, Springer Nature, 2022.
- Caramia M., Giordani S., Guerriero F., Musmanno R., Pacciarelli D., “Ricerca Operativa”, Isedi, 2014, Novara.



UNIVERSITÀ DELLA CALABRIA



Introduzione alle funzionalità del pacchetto

IBM ILOG CPLEX Optimization Studio

A cura di:

Cristian Belfiore, Dottore Magistrale in Ingegneria Gestionale

de-Health Lab - Laboratory of Decision Engineering for Health Care Services

Ponte Pietro Bucci 41C, 8° piano - 87036 Rende (Cosenza)

mail: cristian.belfiore@unical.it



UNIVERSITÀ DELLA CALABRIA 
DIPARTIMENTO DI
INGEGNERIA MECCANICA,
ENERGETICA E GESTIONALE
DIMEG